Prisoner’s game

```
* Tommaso Sgarbanti - 1185058

************* MODELLAZIONE DEL PROBLEMA CON 5 PRIGIONIERI ************

LIGHTon = 'enterON.('exit.LIGHTon + turnOFF.'exit.(continue.LIGHToff +
finish.END));
LIGHToff = 'enterOFF.('exit.LIGHToff + turnON.'exit.LIGHTon);
END = 0;

Pcounter0 = enterON.'turnOFF.exit.'continue.Pcounter1 +
enterOFF.exit.Pcounter0;
Pcounter1 = enterON.'turnOFF.exit.'continue.Pcounter2 +
enterOFF.exit.Pcounter1;
Pcounter2 = enterON.'turnOFF.exit.'continue.Pcounter3 +
enterOFF.exit.Pcounter2;
Pcounter3 = enterON.'turnOFF.exit.'continue.Pcounter4 +
enterOFF.exit.Pcounter3;
Pcounter4 = enterON.'turnOFF.exit.'continue.Pcounter5 +
enterOFF.exit.Pcounter4;
Pcounter5 = enterON.'turnOFF.exit.'continue.Pcounter6 +
enterOFF.exit.Pcounter5;
Pcounter6 = enterON.'turnOFF.exit.'continue.Pcounter7 +
enterOFF.exit.Pcounter6;
Pcounter7 = enterON.'turnOFF.exit.'finish.0 + enterOFF.exit.Pcounter7;

Ptic0 = enterON.exit.Ptic0 + enterOFF.'turnON.exit.Ptic1;
Ptic1 = enterON.exit.Ptic1 + enterOFF.'turnON.exit.Ptic2;
Ptic2 = enterON.exit.Ptic2 + enterOFF.exit.Ptic2;

LIGHT = LIGHTon + LIGHToff;
PRISONERS = Pcounter0 | Ptic0 | Ptic0 | Ptic0 | Ptic0;

set L = { enterON, enterOFF, exit, turnON, turnOFF, continue, finish };

PrisonerGameSuccess = (PRISONERS | LIGHT) \ L;


******************** VERIFICHE DEL COMPORTAMENTO ********************

*** La sfida finisce dopo 2*(n-1) volte che è stata spenta la luce ***
PcounterSpec0 = enterON.'turnOFF.'off.exit.'continue.PcounterSpec1 +
enterOFF.exit.PcounterSpec0;
PcounterSpec1 = enterON.'turnOFF.'off.exit.'continue.PcounterSpec2 +
enterOFF.exit.PcounterSpec1;
PcounterSpec2 = enterON.'turnOFF.'off.exit.'continue.PcounterSpec3 +
enterOFF.exit.PcounterSpec2;
PcounterSpec3 = enterON.'turnOFF.'off.exit.'continue.PcounterSpec4 +
enterOFF.exit.PcounterSpec3;
PcounterSpec4 = enterON.'turnOFF.'off.exit.'continue.PcounterSpec5 +
enterOFF.exit.PcounterSpec4;
```

```
PcounterSpec5 = enterON.'turnOFF.'off.exit.'continue.PcounterSpec6 +
enterOFF.exit.PcounterSpec5;
PcounterSpec6 = enterON.'turnOFF.'off.exit.'continue.PcounterSpec7 +
enterOFF.exit.PcounterSpec6;
PcounterSpec7 = enterON.'turnOFF.'off.exit.'finish.0 +
enterOFF.exit.PcounterSpec7;

ENDSpec = 'end.0;
LIGHTonSpec = 'enterON.('exit.LIGHTonSpec + turnOFF.'exit.
(continue.LIGHToffSpec + finish.ENDSpec));
LIGHToffSpec = 'enterOFF.('exit.LIGHToffSpec + turnON.'exit.LIGHTonSpec);
LIGHTSpec = LIGHTonSpec + LIGHToffSpec;
PRISONERSSpec = PcounterSpec0 | Ptic0 | Ptic0 | Ptic0 | Ptic0;

PrisonerGameSuccessSpec = (PRISONERSSpec | LIGHTSpec) \ L;

Spec = 'off.'off.'off.'off.'off.'off.'off.'off.'end.0;

** Termina dopo 2*(n-1) /2*(n-1)-1/ volte che è stata accesa la luce *
PticSpec0 = enterON.exit.PticSpec0 + enterOFF.'turnON.'on.exit.PticSpec1;
PticSpec1 = enterON.exit.PticSpec1 + enterOFF.'turnON.'on.exit.Ptic2;

PRISONERSSpec_ = Pcounter0 | PticSpec0 | PticSpec0 | PticSpec0 | PticSpec0;

PrisonerGameSuccessSpecON = (PRISONERSSpec_ | LIGHTonSpec) \ L;
PrisonerGameSuccessSpecOFF = (PRISONERSSpec_ | LIGHToffSpec) \ L;

SpecON = 'on.'on.'on.'on.'on.'on.'on.'end.0;
SpecOFF = 'on.'on.'on.'on.'on.'on.'on.'on.'end.0;

********** Solo un prigioniero alla volta accede alla stanza *********
PcounterHolder0 =
enterON.'busy.'turnOFF.'free.exit.'continue.PcounterHolder1 +
enterOFF.'busy.'free.exit.PcounterHolder0;
PcounterHolder1 =
enterON.'busy.'turnOFF.'free.exit.'continue.PcounterHolder2 +
enterOFF.'busy.'free.exit.PcounterHolder1;
PcounterHolder2 =
enterON.'busy.'turnOFF.'free.exit.'continue.PcounterHolder3 +
enterOFF.'busy.'free.exit.PcounterHolder2;
PcounterHolder3 =
enterON.'busy.'turnOFF.'free.exit.'continue.PcounterHolder4 +
enterOFF.'busy.'free.exit.PcounterHolder3;
PcounterHolder4 =
enterON.'busy.'turnOFF.'free.exit.'continue.PcounterHolder5 +
enterOFF.'busy.'free.exit.PcounterHolder4;
PcounterHolder5 =
enterON.'busy.'turnOFF.'free.exit.'continue.PcounterHolder6 +
enterOFF.'busy.'free.exit.PcounterHolder5;
PcounterHolder6 =
enterON.'busy.'turnOFF.'free.exit.'continue.PcounterHolder7 +
enterOFF.'busy.'free.exit.PcounterHolder6;
PcounterHolder7 = enterON.'busy.'turnOFF.'free.exit.'finish.0 +
enterOFF.'busy.'free.exit.PcounterHolder7;
```

```
PticHolder0 = enterON.'busy.'free.exit.PticHolder0 +
enterOFF.'busy.'turnON.'free.exit.PticHolder1;
PticHolder1 = enterON.'busy.'free.exit.PticHolder1 +
enterOFF.'busy.'turnON.'free.exit.PticHolder2;
PticHolder2 = enterON.'busy.'free.exit.PticHolder2 +
enterOFF.'busy.'free.exit.PticHolder2;

PRISONERSHolder = PcounterHolder0 | PticHolder0 | PticHolder0 | PticHolder0
| PticHolder0;

PrisonerGameSuccessME = (PRISONERSHolder | LIGHT) \ L;

* Considerando solo schedulazioni fair, prima o poi la sfida finisce *
PrisonerGameSuccessENDS = (PRISONERS | LIGHTSpec) \ L;

************** Al termine della sfida la stanza è libera *************
PrisonerGameSuccessFREE = (PRISONERSHolder | LIGHTSpec) \ L;
```

Satisfied	375 ms	PrisonerGameSuccessSpec ≈ Spec
Satisfied	250 ms	PrisonerGameSuccessSpecON ≈ SpecON
Satisfied	275 ms	PrisonerGameSuccessSpecOFF ≈ SpecOFF
Satisfied	450 ms	PrisonerGameSuccessME ⊨ ME NoBusyUFree min= (<'free>T and [['busy]]F and ['free][tau]['busy]F) or (['busy]F and <->T and [-]NoBusyUFree) ME max= ((<'busy>T and [tau]F and ['free]F and ['busy]NoBusyUFree) or ['busy]F) and [-]ME
Satisfied	125 ms	PrisonerGameSuccess ⊨ Livelock LOOP max= <tau>LOOP Livelock min= LOOP or <->Livelock
Satisfied	175 ms	PrisonerGameSuccessENDS ⊨ ENDandSTOP ENDandSTOP max= ((<'end>T and [tau]F and <'end>[-]F) or (<tau>T and ['end]F)) and [tau]ENDandSTOP
Satisfied	800 ms	PrisonerGameSuccessENDS ⊨ ENDreachability ENDreachability max= <<'end>>T and [tau]ENDreachability
Satisfied	350 ms	PrisonerGameSuccessFREE ⊨ FreeRoom FreeRoom max= ((<'free><tau><tau><'end>T and ['free][tau][tau]<'end>T and ['free][['busy]]F) or [-][-][-]['end]F) and [-]FreeRoom